void main()
{
    long v, e, r;
    rdVals(v, e, r);

    Graph[][] g = new Graph[][](v);
    foreach (i; 0 .. e)
    {
        long s, t, d;
        rdVals(s, t, d);
        g[s] ~= Graph(t, d);
    }

    long[] Dijkstra(long x)
    {
        long[] result = new long[v];
        result[] = inf;
        result[x] = 0;

        auto que = new BinaryHeap!(Array!Graph, "a.cost > b.cost");
        que.insert(Graph(x, 0));
        while (!que.empty)
        {
            Graph f = que.front;
            que.popFront;
            if (f.cost > result[f.to]) continue;
            foreach (y; g[f.to])
            {
                if (result[y.to] > result[f.to] + y.cost)
                {
                    result[y.to] = result[f.to] + y.cost;
                    que.insert(Graph(y.to, result[y.to]));
                }
            }
        }

        return result;
    }

    long[] dist = Dijkstra(r);
    foreach (x; dist)
    {
        if (x == inf) "INF".writeln;
        else x.writeln;
    }
}

struct Graph
{
    long to, cost;
}

enum long mod = 10^^9 + 7;
enum long inf = 1L << 60;

T rdElem(T = long)()
if (!is(T == struct))
{
    return readln.chomp.to!T;
}

alias rdStr = rdElem!string;
alias rdDchar = rdElem!(dchar[]);

T rdElem(T)()
if (is(T == struct))
{
    T result;
    string[] input = rdRow!string;
    assert(T.tupleof.length == input.length);
    foreach (i, ref x; result.tupleof)
    {
        x = input[i].to!(typeof(x));
    }
    return result;
}

T[] rdRow(T = long)()
{
    return readln.split.to!(T[]);
}

T[] rdCol(T = long)(long col)
{
    return iota(col).map!(x => rdElem!T).array;
}

T[][] rdMat(T = long)(long col)
{
    return iota(col).map!(x => rdRow!T).array;
}

void rdVals(T...)(ref T data)
{
    string[] input = rdRow!string;
    assert(data.length == input.length);
    foreach (i, ref x; data)
    {
        x = input[i].to!(typeof(x));
    }
}

void wrMat(T = long)(T[][] mat)
{
    foreach (row; mat)
    {
        foreach (j, compo; row)
        {
            compo.write;
            if (j == row.length - 1) writeln;
            else " ".write;
        }
    }
}

import std.stdio;
import std.string;
import std.array;
import std.conv;
import std.algorithm;
import std.range;
import std.math;
import std.numeric;
import std.traits;
import std.container;
import std.functional;
import std.typecons;
import std.ascii;
import std.uni;
import core.bitop;
